/**
 * @(#) CodesProxy.as
 *
 * @date		2012. 09. 19.
 * @version		1.0
 * @author		Park, Jutae
 * @description	 CDP신청 Proxy - Proxy
 * 				CDP 신청 리스트를 조회하고 저장, 수정, 삭제한다.
 */
package com.skt.common_hr.pacdp.model
{
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.skt.common_hr.pacdp.PacdpFacade;
	import com.wcl.dataloader.*;
	
	import mx.collections.ArrayCollection;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;
		
	public class CodesProxy extends CommonProxy
	{
		public static const NAME:String = "CodesProxy";
		
		public var cbList:ArrayCollection;
		public var sinChungName:String;
		public var sinChungYear:String;
		
		
		/**
		 * 생성자
		 * 
		 * @param
		 * @return void
		 */
		public function CodesProxy(data:Object=null)
		{
			super(NAME, data);
		}
		
		
		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 * 
		 * @param pernr 사번
		 * @return void
		 */
		public function load(pernr:String):void
		{
			var param:Object = new Object;
			param["I_PERNR"] = pernr;
			
			loadData("Z_HR_CD01", param, loadResult, loadFault); 
		}
		
		
		
		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 * 
		 * @param
		 * @return void
		 */
		public function loadResult(event:ResultEvent):void
		{
			if(event.result["RETCODE"]==0) {  
				cbList = parseCb(event.result.T_CODE);
				sinChungName = event.result.O_ENAME;
				sinChungYear = event.result.O_CDPYR;
				facade.sendNotification(PacdpFacade.LOAD_CODES_COMPLETE, this);
			}
			else {
				trace( "failed to load certi: (" +event.result["RETCODE"] + ") "+event.result["RETTEXT"]);
				facade.sendNotification(PacdpFacade.LOAD_CODES_FAILED, this);
			}
		}
		
		/**
		 * 서버에서 오류가 발생했을 처리
		 * 
		 * @param
		 * @return void
		 */
		private function loadFault(event:FaultEvent):void
		{
			trace("Fault message in CodesProxy: " + event.fault.faultString);
			facade.sendNotification(PacdpFacade.LOAD_CODES_FAILED, this);
		}
		
		
		
		
		
		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 * 
		 * @param
		 * @return 제증명서 종류 Code list
		 */
		private function parseCb(srcArr:ArrayCollection):ArrayCollection
		{
			var retArr:ArrayCollection = new ArrayCollection();
			retArr.addItem({label:'Track 선택', data:-1});
			for each(var rec:Object in srcArr)
			{
				retArr.addItem({label:rec["TEXT"], data:rec["CODE"]});
			}
			return retArr;
		}
	}
}